


**Wavelength Calibration
for the 8614X Series
Optical Spectrum Analyzers**
Product Note 86140-2



Agilent Technologies

Environmental variations such as air pressure, temperature, and humidity can affect the index of refraction of air in the monochromator of the optical spectrum analyzer (OSA).

This product note will discuss calibration methods used to improve the wavelength accuracy in the Agilent 8614X OSA's over a desired wavelength range, providing an example of an external multipoint wavelength user calibration.

This technique uses a tunable laser source and a multi-wavelength meter to correct for errors due to environmental variations and inherent to monochromator design, yielding a typical measurement accuracy better than ± 10 pm. If using an 8614xA that has been upgraded to 8614xB firmware, typical accuracy will be ± 25 pm.

Note

Many aspects of remotely programming the Agilent optical spectrum analyzers are discussed in Product Note 86140-1 *Remote Programming for the Agilent 86140 Series of Optical Spectrum Analyzers*.

Overview

Wavelength calibration routines improve wavelength accuracy by determining errors and correcting them with offsets, using linear interpolation when necessary. For maximum wavelength accuracy, calibration points spaced a maximum of 10 nm apart are recommended.

Wavelength calibration can be accomplished using the optional internal OSA calibrator, an external source at a single wavelength, or an external source at multiple wavelengths.

These calibration routines should only be performed after the instrument's temperature has been stabilized by a minimum of 1 hour of continuous operation.

Each of these methods will optimize wavelength accuracy near the reference source to compensate for environmental variations that affect the index of refraction of air in the OSA monochromator.

Internal Wavelength Calibration

The optional internal calibrator (1513 to 1540 nm) provides a convenient method for increasing wavelength accuracy when used with an internal Enhanced Wavelength Calibration (EWC) process. The wavelength accuracy of the OSA will be ± 0.2 nm over the full wavelength range of the instrument, with ± 10 pm over 1480 to 1570 nm and ± 25 pm accuracy over 1570 to 1620 nm.

The EWC range can be selected for either the full OSA range of 605 nm to 1670 nm, or the 1270 to 1670 nm telecom range, a smaller span more relevant to telecommunications. EWC must be enabled for the wavelength accuracy specifications to apply in the range selected. Setting the range to FULL will require a longer calibration time for an internal calibration, but will provide enhanced wavelength accuracy over the full range.

Manual method using the internal calibrator

1. Access the EWC setup panel:
System > More System Functions... > Service Menu... > Adv Service Functions > More Adv Service Menu > Enhanced Wvl Cal Setup
2. Enable the function, if necessary, and select the desired calibration range.
3. Clean all connectors and connect the internal calibrator to the OSA input.
4. Access the Wavelength Calibration setup panel:
System > Calibration... > Wavelength Cal Setup...
5. Set the signal source to Calibrator.
6. Select Perform Calibration.

Remote method using the internal calibrator

CALibration:WAVelength:EWC:FUNctIon ON	!Enable enhanced wavelength calibration.
CALibration:WAVelength:EWC:RANGe TELE	!Select telecom (1270-1670) nm range for enhanced wavelength calibration.
CALibration:WAVelength:INTernal:NORMal	!Perform internal wavelength calibration.
	!The internal calibrator must be connected before sending this command.

External Single Wavelength Calibration

Using an external single-point calibration source allows the calibration to be done at a specific wavelength. This single wavelength user calibration can be repeated as often as necessary to correct for environmental variations, and existing multipoint wavelength offsets will be adjusted accordingly. After a single wavelength calibration, wavelength accuracy will be ± 10 pm within 10 nm of the reference signal.

The Enhanced Wavelength Calibration (EWC) process can also be used to increase the accuracy of the single-point calibration.

Manual method using an external source

1. Connect the external source to the OSA input.
2. Auto align the OSA to the input signal.
3. Access the Wavelength Calibration setup panel:
System > Calibration... > Wavelength Cal Setup...
4. Select Air or Vacuum reference for the signal source.
5. Set the signal source to External.
6. Select the desired Calibration Wavelength. This wavelength must be within ± 2.5 nm of the source wavelength.
7. Select Perform Calibration.

Remote method using an external source

- For a source with a single peak:

CALibration:WAVelength:VALue <param>	!Set calibration wavelength
CALibration:WAVelength	!Calibrate signal at wavelength

- For a source with multiple peaks:

CALibration:WAVelength:VALue <param>	!Set calibration wavelength
CALCulate:MARKer[1 2 3 4]:X:WAVelength <param>	!Set marker wavelength
CALibration:WAVelength:MARKer	!Calibrate signal at marker

External Multipoint Wavelength Calibration

An external multipoint wavelength calibration can be performed over any specified wavelength range, up to and including the full wavelength range of the OSA (600 nm to 1700 nm). Narrow measurement spans can be chosen to provide greater accuracy over a selected range. Calibrating the wavelength every 10 nm within the desired wavelength range is usually sufficient to improve wavelength accuracy. After a multipoint wavelength calibration, wavelength accuracy will be ± 10 pm within 10 nm of each calibration wavelength. If using an 8614xA that has been upgraded to 8614xB firmware, typical accuracy will be ± 25 pm.

Using the following remote procedure, a signal is sent from a tunable laser source into a multi-wavelength meter and the OSA simultaneously. After measuring the wavelength of the input signal on both instruments, the two values are compared. Taking the multi-wavelength meter readings as actual, the software calculates the error offsets at each wavelength using the equation:

$$WL\ Error = (OSA\ indicated\ WL) - (multi\text{-}wavelength\ meter\ actual\ WL)$$

This procedure is repeated over the entire wavelength range. The data is averaged over narrow wavelength spans to provide a suitable correction for each span. The example below demonstrates this technique.

Once the instrument is calibrated, the new wavelength accuracy can be maintained for many hours without recalibration, assuming a stable temperature environment.

Tip: If the OSA is turned off, the multipoint data will be retained at the next power-on, but the internal thermal shift can introduce inaccuracies to the calibration data. To help compensate for this, a single point calibration using the Offset feature in the Wavelength Calibration Setup panel can be used to adjust the multipoint data. Access this feature by selecting System > Calibration... > Wavelength Cal Setup... and choosing the Offset option before running the single point calibration. To insure this offset process has provided sufficient accuracy, the wavelength readings of the multi-wavelength meter and the OSA should be compared to verify the wavelength accuracy and determine if a full multipoint wavelength recalibration is necessary.

Example

In this example, the sampling is done over 2 nm spans using a tunable laser source stepped every 100 pm and measured by the OSA and the multi-wavelength meter. These spans are taken every 10 nm over the desired calibration range. For example, if you are measuring from 1500 nm to 1600 nm, you might sample from 1509-1511 nm in 100 pm steps, then move to 1519-1521 nm, and so on. Sampling over a 100 nm span with these parameters usually takes about 25 minutes.

Each 2 nm span generates a set of points. For each such set, a wavelength offset pair is determined. The average of the minimum error and the maximum error is calculated to determine the necessary offset. This average is applied to the wavelength located at the midpoint of the two extremes.

Following is a sample set of wavelength offsets:

Wavelength	Offset	Wavelength	Offset
1509.0 nm	12 pm	1510.0 nm	14 pm
1509.1 nm	17 pm	1510.1 nm	6 pm
1509.2 nm	13 pm	1510.2 nm	14 pm
1509.3 nm	15 pm	1510.3 nm	16 pm
1509.4 nm	5 pm*	1510.5 nm	12 pm
1509.5 nm	11 pm	1510.6 nm	9 pm
1509.6 nm	9 pm	1510.7 nm	11 pm
1509.7 nm	17 pm	1510.8 nm	15 pm
1509.8 nm	19 pm**	1510.9 nm	8 pm
1509.9 nm	10 pm	1511.0 nm	11 pm

* Minimum offset

** Maximum offset

This indicates a maximum offset of 19 pm at 1509.8 nm, and a minimum offset of 5 pm at 1509.4 nm for an average offset of 12 pm. The average wavelength between these two points is 1509.6 nm. This gives us the wavelength offset pair (1509.6e-9, 12e-12) for the measurements taken around 1510 nm.

This calculation is performed for each span in the calibration range, then all of these points are concatenated into a comma delimited string for entry into the OSA. The calibration string sent to the OSA must contain an even number of values, taking the form <wavelength 1, offset 1, wavelength 2, offset 2, ..., wavelength n, offset n>. For example, (1509.6e-9, 12e-12) and (1520e-9, 26.4e-12) would become the string <1509.6e-9, 12e-12, 1520e-9, 26.4e-12>.

Once this string is input to the OSA, the user calibration is complete and the OSA will maintain improved wavelength accuracy. It is recommended that the wavelength readings of the multi-wavelength meter and the OSA be compared periodically to verify the wavelength accuracy of the calibration. In this manner, over time, the need to recalibrate the OSA can be determined.

Note

The resulting wavelengths in the correction string can be no closer than 2 pm and must be input with units in meters. The maximum offset that can be entered is ± 200 pm.

Procedure

Required Equipment:

- Personal computer or workstation equipped with a GPIB (also known as IEEE-488 interface bus) card and instrument control software.
- Tunable laser source (TLS).
- High accuracy multi-wavelength meter.
- Fiber optic coupler.
- 3 - fiber optic patch cords to connect the instruments to the tunable laser source.
- 3 - GPIB cables to connect each of the instruments to the PC.
- OSA firmware revision B.01.00 or greater. Upgrading firmware on 8614xA OSA's will enable you to perform multipoint wavelength calibrations. More information on upgrading can be found at www.agilent.com/comms/osaupgrade.




Figure 1. Test setup of the OSA, PC, tunable laser source, multi-wavelength meter and cables

1. Write a program in a language appropriate for your test station. The following programming example in HP Basic has been provided as a guide to the sequence of commands necessary to perform an external wavelength calibration.
2. Connect the equipment as in Figure 1. This example uses an Agilent 86120C multi-wavelength meter and an Agilent 8164A/81640A tunable laser source.
3. After warm-up and environmental stabilization, and just before beginning the multipoint calibration, perform a manual, external wavelength calibration using the Replaced option in the Wavelength Calibration Setup panel. If there is no existing multipoint data to replace, this option will be grayed out. The TLS can be used as the source and the multi-wavelength meter will indicate the external calibration wavelength value.

4. The program sets up and runs the calibration as follows:

a. The OSA is placed into a high performance state, setting the following parameters:

Resolution bandwidth - 0.06 nm for the 86142 or 86145 OSA,
0.07 nm for the 86140, 86141 or 86143 OSA.

Video bandwidth - narrowest possible for your setup.

Sweep time - Auto.

Wavelength span - as appropriate for your setup.

Trace points - as appropriate for your setup.

b. The auto-align routine is run on the OSA.

c. Any existing multipoint calibration data is disabled so it does not interfere with collecting new data.

d. The tunable laser source is stepped from wavelength-to-wavelength.

e. At each point, the wavelength is measured by both the multi-wavelength meter and the OSA. The output power and settling time of the tunable laser source varies from instrument to instrument so the multi-wavelength meter reading is taken both before and after the OSA reading to verify the tunable laser source stability within ± 1 pm.

f. Various checks are performed to insure valid data:

- To verify a real signal is present, the 3 dB bandwidth of the signal at the OSA must be less than full span and the peak amplitude must be above -70 dBm.
- To verify the minimum 2 pm spacing required for valid correction data, the TLS wavelength steps are checked as measured by the multi-wavelength meter.
- The slope of the correction data is checked to be less than 1 and greater than -1. For example, a correction of -8 pm at 1550 nm, followed by a correction of +100 pm at 1550.1 nm would be a slope of 108pm/100pm which is not allowed.
- The magnitude of the correction is checked to be less than 200 pm.

g. The values are compared and the wavelength errors calculated.

h. A correction string is generated from this data.

i. The correction string is sent to the OSA and applied to future wavelength measurements.

Remote Commands

The Agilent 86140B Series Optical Spectrum Analyzer User's Guide provides detailed information on remote programming of the instrument. Only commands specific to this application are presented here.

```
CALibration:WAVeLength:EWC:FUNCTion ON|OFF|0|1
CALibration:WAVeLength:EWC:FUNCTion?
```

Enables or disables the enhanced wavelength calibration for subsequent calibrations. EWC must be enabled for wavelength accuracy specifications to apply in the range selected.

0 = disables EWC

1 = enables EWC (default on factory preset)

```
CALibration:WAVeLength:EWC:RANGe FULL|TELEcom
CALibration:WAVeLength:EWC:RANGe?
```

Sets the range over which the enhanced wavelength calibration (EWC) is performed. The two ranges for the EWC are FULL and TELEcom. FULL covers the range from 605 nm to 1670 nm. TELEcom covers the smaller span more relevant to telecommunications: 1270 to 1670 nm. Factory preset is TELEcom.

When enabled, the EWC is applied during internal calibrations. EWC must be enabled for wavelength accuracy specifications to apply in the range selected. Setting the range to FULL will require a longer calibration time for an internal calibration, but will provide enhanced wavelength accuracy over the full range.

```
CALibration:WAVeLength[:EXTernal]:MULTipoint
```

Performs a single point enhanced wavelength calibration using an external source. Adjusts the multipoint data at the wavelength selected by the CALibration:WAVeLength:EXTernal:VALue command. If the wavelength measured on the input signal differs more than ± 2.5 nm from the value specified, the calibration is aborted.

Note

For this command to function properly, it must be used in the correct sequence with the following commands:

```
CALibration:WAVeLength:EXTernal:VALue <param>
```

```
CALibration:WAVeLength[:EXTernal]:MULTipoint
```


`CALibration:WAVelength[:EXternal]:MULTipoint:MARKer[1|2|3|4|]`

Performs a single point enhanced wavelength calibration using the signal nearest the marker. The marker location must be selected before this command can be run. Adjusts the multipoint data at the wavelength selected by the `CALibration:WAVelength:EXternal:VALue` command. If the wavelength measured on the input signal differs more than ± 2.5 nm from the value specified, the calibration is aborted. If no multipoint data exists, the calibration is aborted and a settings conflict error is generated.

This command is necessary if a signal with two or more peaks is input to the optical spectrum analyzer during the calibration. If a source has more than one peak, the marker is used to determine which peak will be calibrated.

Note

For this command to function properly, it must be used in the correct sequence with the following commands:

`CALibration:WAVelength:EXternal:VALue <param>`

`CALCulate:MARKer[1|2|3|4]:X:WAVelength <param>`

`CALibration:WAVelength[:EXternal]:MULTipoint:MARKer[1|2|3|4]`

`CALibration:WAVelength[:EXternal][:NORMAL]`

Performs a single point enhanced wavelength calibration using an external source. Disables all multipoint wavelength calibration offsets. The multipoint data can also be disabled with `CALibration:WAVelength:MODE:NORMAL`.

If the wavelength measured on the input signal differs more than ± 2.5 nm from the value specified, the calibration is aborted.

Note

For this command to function properly, it must be used in the correct sequence with the following commands:

`CALibration:WAVelength:EXternal:VALue <param>`

`CALibration:WAVelength[:EXternal][:NORMAL]`

CALibration:WAVelength[:EXTErnal][:NORMal]:MARKer[1|2|3|4|]

Performs a single point enhanced wavelength calibration using the signal nearest the marker. The marker location must be selected before this command can be run. Disables all multipoint wavelength calibration offsets. The multipoint data can also be disabled with CALibration:WAVelength:MODE:NORMal. If the wavelength measured on the input signal differs more than ± 2.5 nm from the value specified in the CALibration:WAVelength:VALue command, the calibration is aborted.

This command is necessary if a signal with two or more peaks is input to the optical spectrum analyzer during the calibration. If a source has more than one peak, the marker is used to determine which peak will be calibrated.

Note

For this command to function properly, it must be used in the correct sequence with the following commands:

CALibration:WAVelength:EXTErnal:VALue <param>

CALCulate:MARKer[1|2|3|4]:X:WAVelength <param>

CALibration:WAVelength[:EXTErnal][:NORMal]:MARKer[1|2|3|4]

CALibration:WAVelength[:EXTErnal]:VALue <param> [M|UM|NM|A]

CALibration:WAVelength[:EXTErnal]:VALue?

Specifies the wavelength for a single point calibration. Default units for the parameter are meters.

CALibration:WAVelength:INTernal:MULTipoint

Performs an enhanced wavelength calibration using the internal calibrator. Any existing multipoint wavelength calibration data is adjusted relative to this calibration. If no multipoint data exists, the calibration is aborted and a settings conflict error is generated.

Note

The internal calibrator must be connected to the input before sending this command.

CALibration:WAVelength:INTernal[:NORMal]

Performs an enhanced wavelength calibration using the internal calibrator. Any existing multipoint wavelength calibration data is cleared.

Note

The internal calibrator must be connected to the input before sending this command.

CALibration:WAVelength:MODE NORMal|MULTipoint
CALibration:WAVelength:MODE?

Enables or disables the multipoint wavelength calibration data. NORMal disables the multipoint wavelength calibration data. MULTipoint enables the data from the last multipoint wavelength calibration per CALibration:WAVelength:MULTipoint:DATA. This data must be entered before MULTipoint mode can be selected.

The following commands change the setting of
CALibration:WAVelength:MODE to NORMal:

CALibration:WAVelength[:EXternal]:NORMal
CALibration:WAVelength[:EXternal]:NORMal:MARKer[1|2|3|4]

Once multipoint data is entered, the following commands will enable the multipoint data. Refer to the specific commands for further information.

CALibration:WAVelength[:EXternal]:MULTipoint
CALibration:WAVelength[:EXternal]:MULTipoint:MARKer[1|2|3|4]
CALibration:WAVelength:INTernal:MULTipoint
CALibration:WAVelength:MULTipoint:DATA

CALibration:WAVelength:MULTipoint:DATA X1,Y1,X2,Y3,...,Xn,Yn
CALibration:WAVelength:MULTipoint:DATA?

Enters user measured external multipoint calibration data. The command takes the data in <string> format and writes it to the wavelength calibration table.

Xn are wavelengths in vacuum in meters of the wavelength standard (not the value indicated by the OSA). The Xn minimum spacing is 2 pm, and must be in increasing order. There is a maximum of 10000 pairs. Linear interpolation is used between the data points when calculating the wavelength corrections.

Y_n are wavelength errors in vacuum (indicated wavelength - actual wavelength) in meters. Y_n magnitude must be less than 200 pm.

The spacing between data points must be larger than the magnitude of the change in error between data points. Specifically, the magnitude of the slope must be less than 1. Where $\text{slope} = (Y(n+1) - Y(n)) / (X(n+1) - X(n))$. For example if X_n are 10 pm apart, Y_n must change by less than 10 pm.

The query returns any external multipoint wavelength calibration data in string format. For example:

```
+1.45011471E-006,+0.00000000E+000,+1.50011168E-006,+9.20199449E-13,  
+1.56010779E-006,-1.12468277E-012,+1.61010432E-006,+0.00000000E+000
```

Previous multipoint wavelength data are cleared each time the command is used. Therefore, to modify the multipoint wavelength calibration data, use the query to obtain the existing table of data, then make changes to the table and reenter it using this command.

When measuring new external multipoint calibration data, use "CALibration:WAVelength:MODE:NORMal" to disable previous wavelength calibration data.

CALibration:WAVelength:MULTipoint:DELeTe

Deletes calibration data entered by
CALibration:WAVelength:MULTipoint:DATA.

CALibration:WAVelength:USER:DATA <string>
CALibration:WAVelength:USER:DATA?

Although this command is available, some OSA firmware versions do not support it. In place of this command, it is recommended that you use: CALibration:WAVelength:MULTipoint:DATA.

All information given for CALibration:WAVelength:MULTipoint:DATA will apply to this command.

Sample Program

The following is a sample user calibration program written in HP BASIC for Windows using the Agilent 8168 tunable laser source and the 86120C multi-wavelength meter.

```

10 !INITIALIZE VARIABLES
20 !Variable definition: This example sets up a calibration point every 10 nm (Cal_inc),
30 !starting at 1530 nm (Start_wl) and ending at 1560 nm (Stop_wl). Wavelength offsets are
40 !measured over a 2nm range (Cal_span) taken every 0.1 nm (Cal_wl_inc) centered at the
50 !calibration wavelengths (1530, 1540, 1550, and 1560). The maximum and minimum offsets
60 !are then averaged, and the result is entered as the offset for the center wavelength.
70 !
80 !This example program will also set a zero offset at the wavelengths +Cal_inc from the
90 !Stop_wl, and -Cal_inc from the Start_wl, if Cal_inc is at least 0.2nm. If Cal_inc is less than 0.2nm,
100 !a zero will be entered 0.2nm before the start and after the stop wavelengths.
110 !For this example, a zero offset is entered at 1520 and 1570nm. The OSA interpolates
120 !offset wavelengths between those entered in the calibration procedure. Inserting zeroes
130 !at either end of the calibration string ensures that offsets are zeroed outside the calibration region.
140 !
150 !*****Program Start*****
160 !
170 !*****Global Variable Declaration*****
180 !
190 Start_wl=1530                                !Start WL, nm
200 Stop_wl=1560                                !Stop WL, nm
210 Cal_inc=10                                  !Calibration increment, nm
220 Cal_span=2                                  !Calibration span, nm
230 Sweep_span=.4                               !Sweep span used when taking data, nm
240 Cal_wl_inc=0.1                              !Calibration wavelength increment, nm
250 DIM WL_cal_string$(32767)                  !Initialize string to store cal offsets
260 DIM New_cal_string$(32767)                 !Initialize temporary string
270 Infinity=999999999                         !Variable for maximum offset
280 Not_a_number=9.91E+37                      !SCPI definition for undefined values
290 Offset=0                                    !Initialize offset
300 Offset_wl=(Start_wl-(Cal_span/2)-.2)*E-9    !Initialize offset wl for slope check
310 !
320 !*****
330 !The following check ensures that the calibration points are not within 2 pm of one another.
340 !This is the lower limit for calibration point spacing in the OSA
350 !*****
360 !
370 IF Cal_span>(Cal_inc-.002) THEN
380   PRINT "Cal_inc must be at least 2 pm larger than Cal_span. Stopping Program"
390   GOTO 2150                                !Go to end of program
400 END IF
410 !
420 !
430 !*****Initialize Multiwavelength Meter*****
440 CLEAR 720
450 ASSIGN @Mwm TO 720
460 OUTPUT @Mwm;"*RST"                        !Reset multiwavelength meter
470 OUTPUT @Mwm;"SENS:CORR:MED VAC"           !Display WL in Vacuum
480 !
490 !*****Initialize Tunable Laser Source*****

```

```

500 CLEAR 724
510 ASSIGN @Tls TO 724
520 OUTPUT @Tls;"*RST"                                !Reset TLS
530 OUTPUT @Tls;"POW:UNIT DBM"                          !Set power units to DBM
540 OUTPUT @Tls;"POW -11DBM"                            !Set output power to -11DBM
550 OUTPUT @Tls;"OUTP ON"                                !Turn output on
560 OUTPUT @Tls;"SOUR:WAV "&VAL$(Start_wl)&"nm"          !Set output wl to mid range
570 !
580 !*****Initialize OSA and Set to High Performance State*****
590 CLEAR 723
600 ASSIGN @Osa TO 723;EOL CHR$(10) END                !Set terminating character to LF w/ EOI
610 OUTPUT @Osa;"*RST"                                !Reset OSA
620 OUTPUT @Osa;"SWE:POIN 401"                          !Set # of trace points to 401
630 OUTPUT @Osa;"SENS:CORR:RVEL:MED VAC"                !Display WL in VAC
640 OUTPUT @Osa;"SENS:WAV:SPAN "&VAL$(Sweep_span)&"NM"    !Set span
650 OUTPUT @Osa;"SWE:TIME:AUTO ON"                      !Set sweep time to auto
660 OUTPUT @Osa;"SENS:BAND:VID 194HZ"                  !Set video bandwidth to 194HZ
670 OUTPUT @Osa;"SENS:BAND 0.06NM"                    !Set resolution bandwidth to 0.06nm
680 OUTPUT @Osa;"CALC:MARK1:TRAC TRA"                  !Marker on trace A
690 OUTPUT @Osa;"CALC:MARK1:FUNC:BAND ON"              !Turn on bandwidth marker
700 OUTPUT @Osa;"DISP:WIND:TRAC:Y:SCAL:RLEV -20DBM"     !Set reference level to -20DBM
710 OUTPUT @Osa;"CAL:WAV:MODE NORM"                   !Turn off any existing multipoint data
720 OUTPUT @Osa;"SENS:WAV:CENT "&VAL$(Start_wl)&"NM"      !Set center wl for auto align
730 OUTPUT @Osa;"INIT:IMM"                             !Take a sweep
740 OUTPUT @Osa;"CALC:MARK1:MAX"                       !Mark Peak WL
750 OUTPUT @Osa;"CAL:ALIG:MARK1"                       !Perform Auto Align on TLS Signal
760 !
770 !*****Measurement Loops*****
780 !
790 !*****Outer FOR Loop*****
800 !Steps the calibration wavelength (Cal_wl) from the start wavelength (Start_wl) to the stop
810 ! wavelength (Stop_wl) in increments of the calibration increment (Cal_inc).
820 !
830 FOR Cal_wl=Start_wl TO Stop_wl STEP Cal_inc
840 !Variable declaration
850 Max_offset=-Infinity                                !Initialize maximum offset
860 Min_offset=Infinity                                  !Initialize minimum offset
870 Max_wl=0                                             !Set wl of maximum offset to zero
880 Min_wl=0                                             !Set wl of minimum offset to zero
890 !
900 !*****Inner FOR Loop*****
910 !Sets the wavelength to be measured in steps of the calibration wavelength (Cal_wl_inc) for the
920 !calibration span (cal_span) around the calibration wavelength (Cal_wl).
930 !
940 !
950 FOR Current_wl=Cal_wl-Cal_span/2 TO Cal_wl+Cal_span/2 STEP Cal_wl_inc
960 !
970 !*****Set TLS Output Wavelength*****
980 !
990 OUTPUT @Tls;"SOUR:WAVE "&VAL$(Current_wl)&"NM"        !Set TLS output to current wl
1000 OUTPUT @Tls;"*OPC?"                                !Wait for TLS to settle
1010 ENTER @Tls;Done
1020 !
1030 !*****Initialize Variables for Mode-Hop Check*****
1040 Mwm_wavelength1=-Infinity
1050 Mwm_wavelength2=Infinity

```

```

1060 !
1070 !*****Mode-Hop Check Loop*****
1080 !The following loop ensures that the laser is not mode-hopping, by checking
1090 !the laser wavelength with the multiwavelength meter before and after the
1100 !OSA measurement. The two multiwavelength meter readings must agree within
1110 !1.0 pm for the reading to be accepted.
1120 !
1130 WHILE (ABS(Mwm_wavelength1-Mwm_wavelength2)>1.E-12)
1140 !
1150 !*****First Multiwavelength Meter Read*****
1160 OUTPUT @Mwm;"INIT:IMM;*OPC?" !Take reading and wait to complete
1170 ENTER @Mwm;Done
1180 OUTPUT @Mwm;"FETC:SCAL:POW:WAV?" !Query maximum wavelength
1190 ENTER @Mwm;Mwm_wavelength1
1200 !
1210 !*****Read with OSA*****
1220 !
1230 OUTPUT @Osa;"WAV:CEN "&VAL$(Current_wl)&"NM" !Set center wl to the current wl
1240 OUTPUT @Osa;"INIT:IMM" !Take sweep
1250 OUTPUT @Osa;"CALC:MARK1:MAX" !Mark peak wavelength
1260 OUTPUT @Osa;"CALC:MARK1:FUNC:BAND:RES?" !Query BW 3dB points
1270 ENTER @Osa;Osa_bw
1280 OUTPUT @Osa;"CALC:MARK1:Y?" !Query peak amplitude
1290 ENTER @Osa;Osa_peak
1300 IF ((Osa_bw<Not_a_number) AND (Osa_peak>-70)) THEN
1310     OUTPUT @Osa;"CALC:MARK1:FUNC:BAND:X:CEN?" !Query wl at mean of 3dB points
1320     ENTER @Osa;Osa_wavelength !Store value as Osa_wavelength
1330 ELSE
1340     PRINT "Signal not found at "&VAL$(Current_wl)&" nm"
1350     GOTO 1600
1360 END IF
1370 !
1380 !*****Second Multiwavelength Meter Read*****
1390 OUTPUT @Mwm;"INIT:IMM;*OPC?" !Take reading and wait to complete
1400 ENTER @Mwm;Done
1410 OUTPUT @Mwm;"FETC:SCAL:POW:WAV?" !Query maximum wavelength
1420 ENTER @Mwm;Mwm_wavelength2
1430 !
1440 END WHILE !End of mode-hop check loop
1450 !
1460 !*****Find the Wavelength Calibration Offset for the OSA at the Current Wavelength*****
1470 Mwm_wavelength=(Mwm_wavelength1+Mwm_wavelength2)/2
1480 Difference=Osa_wavelength-Mwm_wavelength
1490 !
1500 !*****Update Maximum and Minimum Offsets Within Calibration Span*****
1510 IF Difference>Max_offset THEN
1520     Max_offset=Difference !Store max offset
1530     Max_wl=Mwm_wavelength !Store wavelength of max offset
1540 END IF
1550 IF Difference<Min_offset THEN
1560     Min_offset=Difference !Store min offset
1570     Min_wl=Mwm_wavelength !Store wavelength of min offset
1580 END IF
1590 !

```

```

1600 NEXT Current_wl                                !End of inner FOR loop
1610 !
1620 !*****Store Previous Offset Pair for Slope Check*****
1630 Last_offset_wl=Offset_wl
1640 Last_offset=Offset
1650 !
1660 !*****Find Offset Pair for Current Calibration Span*****
1670 Offset_wl=(Max_wl+Min_wl)/2
1680 Offset=(Max_offset+Min_offset)/2
1690 !
1700 !*****Check Slope and Offset Against Limits*****
1710 IF (Offset_wl>Last_offset_wl) THEN
1720   Slope=(Offset-Last_offset)/(Offset_wl-Last_offset_wl)
1730 ELSE
1740   Slope=Infinity
1750 END IF
1760 IF ((ABS(Offset)<2.00E-10) AND (ABS(Slope)<=1)) THEN
1770   New_cal_string$=New_cal_string$&VAL$(Offset_wl)&","&VAL$(Offset)&","
1780 ELSE
1790   PRINT "Calibration Point at: "&VAL$(Cal_wl)&" nm correction value unreasonable, point ignored"
1800 END IF
1810 !
1820 NEXT Cal_wl                                    !End of outer FOR loop
1830 !
1840 !*****Build Calibration String*****
1850 !The following builds the calibration string with an initial and a final zero offset from the
1860 !first and final calibration wavelengths for interpolation outside the range. The offsets are
1870 !entered at least 0.2 nm from the neighboring calibration wavelengths to avoid a possible slope
1880 !violation (the largest offset allowed is 0.2 nm). If the Cal_inc spacing is larger than 0.2 nm
1890 !the zeroes are entered into cal_inc from the start and stop wavelengths
1900 !
1910 IF Cal_inc>.2 THEN
1920   WI_cal_string$=VAL$(Start_wl-Cal_inc)&"e-9,0,"          !Initialize String w/ first zero offset
1930   WI_cal_string$=WI_cal_string$&New_cal_string$          !Append new calibration offset data
1940   WI_cal_string$=WI_cal_string$&VAL$(Stop_wl+Cal_inc)&"e-9,0" !Append final zero offset
1950 ELSE
1960   WI_cal_string$=VAL$(Start_wl-.2)&"e-9,0,"              !Initialize string w/ first zero offset
1970   WI_cal_string$=WI_cal_string$&New_cal_string$          !Append new calibration offset data
1980   WI_cal_string$=WI_cal_string$&VAL$(Stop_wl+.2)&"e-9,0" !Append final zero offset
1990 END IF
2000 !
2010 !*****Load Calibration String Into OSA*****
2020 !
2030 OUTPUT @Osa;"CAL:WAV:MULT:DATA ";WI_cal_string$
2040 OUTPUT @Osa;"CAL:WAV:MULT:DATA?"
2050 DIM WI_cal_check$(32767)
2060 ENTER @Osa;WI_cal_check$
2070 PRINT "WAVELENGTH CALIBRATION STRING="&WI_cal_string$
2080 PRINT "OSA CALIBRATION SETTINGS="&WI_cal_check$
2090 !
2100 !*****Cleanup*****
2110 OUTPUT @Tls;"OUTP OFF"                                !Turn off laser
2120 LOCAL @Osa                                           !Release remote control
2130 LOCAL @Mwm
2140 LOCAL @Tls
2150 END

```


Agilent Technologies'

Test and Measurement Support, Services, and Assistance

Agilent Technologies aims to maximize the value you receive, while minimizing your risk and problems. We strive to ensure that you get the test and measurement capabilities you paid for and obtain the support you need. Our extensive support resources and services can help you choose the right Agilent products for your applications and apply them successfully. Every instrument and system we sell has a global warranty. Support is available for at least five years beyond the production life of the product. Two concepts underlie Agilent's overall support policy: "Our Promise" and "Your Advantage."

Our Promise

Our Promise means your Agilent test and measurement equipment will meet its advertised performance and functionality. When you are choosing new equipment, we will help you with product information, including realistic performance specifications and practical recommendations from experienced test engineers. When you use Agilent equipment, we can verify that it works properly, help with product operation, and provide basic measurement assistance for the use of specified capabilities, at no extra cost upon request. Many self-help tools are available.

Your Advantage

Your Advantage means that Agilent offers a wide range of additional expert test and measurement services, which you can purchase according to your unique technical and business needs. Solve problems efficiently and gain a competitive edge by contracting with us for calibration, extra-cost upgrades, out-of-warranty repairs, and on-site education and training, as well as design, system integration, project management, and other professional engineering services. Experienced Agilent engineers and technicians worldwide can help you maximize your productivity, optimize the return on investment of your Agilent instruments and systems, and obtain dependable measurement accuracy for the life of those products.

By internet, phone, or fax, get assistance with all your test & measurement needs.

Online assistance:

www.agilent.com/comms/lightwave

Phone or Fax

United States:

(tel) 1 800 452 4844

Canada:

(tel) 1 877 894 4414

(fax) (905) 282 6495

Europe:

(tel) (31 20) 547 2323

(fax) (31 20) 547 2390

Japan:

(tel) (81) 426 56 7832

(fax) (81) 426 56 7840

Latin America:

(tel) (305) 269 7500

(fax) (305) 269 7599

Australia:

(tel) 1 800 629 485

(fax) (61 3) 9210 5947

New Zealand:

(tel) 0 800 738 378

(fax) 64 4 495 8950

Asia Pacific:

(tel) (852) 3197 7777

(fax) (852) 2506 9284

Product specifications and descriptions in this document subject to change without notice.

Copyright © 2000, 2001 Agilent Technologies

Printed in USA May 15, 2001

5980-0043E



Agilent Technologies